NAG Library Routine Document

F07JUF (ZPTCON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07JUF (ZPTCON) computes the reciprocal condition number of a complex n by n Hermitian positive definite tridiagonal matrix A, using the LDL^{H} factorization returned by F07JRF (ZPTTRF).

2 Specification

```
SUBROUTINE F07JUF (N, D, E, ANORM, RCOND, RWORK, INFO)

INTEGER                 N, INFO
REAL (KIND=nag_wp)      D(*), ANORM, RCOND, RWORK(N)
COMPLEX (KIND=nag_wp)   E(*)
```

The routine may be called by its LAPACK name ***zptcon***.

3 Description

F07JUF (ZPTCON) should be preceded by a call to F07JRF (ZPTTRF), which computes a modified Cholesky factorization of the matrix A as

$$A = LDL^{\mathrm{H}},$$

where L is a unit lower bidiagonal matrix and D is a diagonal matrix, with positive diagonal elements. F07JUF (ZPTCON) then utilizes the factorization to compute $\left\|A^{-1}\right\|_1$ by a direct method, from which the reciprocal of the condition number of A, $1/\kappa(A)$ is computed as

$$1/\kappa_1(A) = 1/\left(\|A\|_1\left\|A^{-1}\right\|_1\right).$$

$1/\kappa(A)$ is returned, rather than $\kappa(A)$, since when A is singular $\kappa(A)$ is infinite.

4 References

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

2: D($*$) – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: must contain the n diagonal elements of the diagonal matrix D from the LDL^{H} factorization of A.

3: E($*$) – COMPLEX (KIND=nag_wp) array *Input*

Note: the dimension of the array E must be at least $\max(1, \mathrm{N}-1)$.

On entry: must contain the $(n-1)$ subdiagonal elements of the unit lower bidiagonal matrix L. (E can also be regarded as the superdiagonal of the unit upper bidiagonal matrix U from the $U^{\mathrm{H}}DU$ factorization of A.)

4: ANORM – REAL (KIND=nag_wp) *Input*

On entry: the 1-norm of the **original** matrix A, which may be computed by calling F06UPF with its parameter NORM = '1'. ANORM must be computed either **before** calling F07JRF (ZPTTRF) or else from a **copy** of the original matrix A.

Constraint: ANORM ≥ 0.0.

5: RCOND – REAL (KIND=nag_wp) *Output*

On exit: the reciprocal condition number, $1/\kappa_1(A) = 1/\left(\|A\|_1 \left\|A^{-1}\right\|_1\right)$.

6: RWORK(N) – REAL (KIND=nag_wp) array *Workspace*

7: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed condition number will be the exact condition number for a closely neighbouring matrix.

8 Further Comments

The condition number estimation requires $O(n)$ floating point operations.

See Section 15.6 of Higham (2002) for further details on computing the condition number of tridiagonal matrices.

The real analogue of this routine is F07JGF (DPTCON).

9 Example

This example computes the condition number of the Hermitian positive definite tridiagonal matrix A given by

$$A = \begin{pmatrix} 16.0 & 16.0 - 16.0i & 0 & 0 \\ 16.0 + 16.0i & 41.0 & 18.0 + 9.0i & 0 \\ 0 & 18.0 - 9.0i & 46.0 & 1.0 + 4.0i \\ 0 & 0 & 1.0 - 4.0i & 21.0 \end{pmatrix}.$$

9.1 Program Text

```
    Program f07jufe

!     F07JUF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x02ajf, zlanht => f06upf, zptcon, zpttrf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: anorm, rcond
      Integer                          :: info, n
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: e(:)
      Real (Kind=nag_wp), Allocatable  :: d(:), rwork(:)
!     .. Executable Statements ..
      Write (nout,*) 'F07JUF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (e(n-1),d(n),rwork(n))

!     Read the lower bidiagonal part of the tridiagonal matrix A from
!     data file

      Read (nin,*) d(1:n)
      Read (nin,*) e(1:n-1)

!     Compute the 1-norm of A
!     f06upf is the NAG name equivalent of the LAPACK auxiliary zlanht
      anorm = zlanht('1-norm',n,d,e)

!     Factorize the tridiagonal matrix A
!     The NAG name equivalent of zpttrf is f07jrf
      Call zpttrf(n,d,e,info)

      If (info==0) Then

!       Estimate the condition number of A
!       The NAG name equivalent of zptcon is f07juf
        Call zptcon(n,d,e,anorm,rcond,rwork,info)

!       Print the estimated condition number

        If (rcond>=x02ajf()) Then
          Write (nout,99999) 'Estimate of condition number = ', &
            1.0E0_nag_wp/rcond
        Else
          Write (nout,99999) 'A is singular to working precision. RCOND = ', &
            rcond
        End If

      Else
        Write (nout,99998) 'The leading minor of order ', info, &
          ' is not positive definite'
      End If

99999 Format (1X,A,1P,E10.2)
99998 Format (1X,A,I3,A)
    End Program f07jufe
```

9.2 Program Data

```
F07JUF Example Program Data
    4                                                       :Value of N
   16.0          41.0          46.0          21.0         :End of diagonal D
 ( 16.0, 16.0) ( 18.0, -9.0) (  1.0, -4.0)                 :End of sub-diagonal E
```

9.3 Program Results

```
 F07JUF Example Program Results

 Estimate of condition number =   9.21E+03
```